
Lab 3: Simple Analyses

Overview

1. Two sample t-test

a. Between coping styles

b. At each time

c. Averaged over the observed baseline values

d. Cases with missing observations deleted

2. Paired t-test: treatment effects
a. Average baselines

i. Overall (Is the average at the 4th time point different from the average of the earlier observations?)
ii. For each treatment by coping style combination
b. Repeat separately by baselines

3. Difference of differences (4th observation minus 2nd observation)

a. By coping style:

i. Compare distract tmt to null tmt

ii. Compare attend tmt to null tmt

4. Random intercept model

Uses all the data.  



Combines features of paired, two-sample and DoD tests.

We go over the basic analyses often perpetrated on longitudinal data, then introduce one of the simplest of longitudinal data models, the random intercept model.  The random intercept model combines aspects of the various simple analyses in one large model that uses all the data available and in an appropriate fashion.  

Lab Tasks

Use the pain data for this project, and the log2 paintol variable as the response.  

*DOWNLOAD PAIN SAS DATA SET;

libname b236 'C:\Documents and Settings\Leanne\Desktop\B236';
1. Two sample t-test

/*NOTE: HOW TO READ THE TTEST OUTPUT.

FIRST NOTE THE EQUALITY OF VARIANCES, Ho: the variances equal at alpha=.05. 


If we fail to reject, then the variances aren’t unequal and we pool them. 


-> LOOK AT POOLED T-TEST


If we reject, then variances are unequal, we cannot pool them


-> LOOK AT SATTHERWAITE T-TEST

REPORT MEAN AND STANDARD ERROR... LOOK AT MEAN COLUMN AND SE COLUMN.

THE CONFIDENCE LIMITS OF THE MEAN AND STANDARD DEVIATION ARE ALSO REPORTED. */
a. Between coping styles at each time 

Which tests are significant at alpha=.05 level?

proc sort data=b236.pain; by trial;run;

proc ttest data=b236.pain;


by trial;


class cs;


var l2paintol;


title 'T-Test by Trial of Log2(Paintol)';


run;
b. Averaged over the observed baseline values

How does this test compare to the previous results?

*TRANSPOSE DATASET TO WIDE FORMAT;
proc sort data=b236.pain; by id;run;

proc transpose  data=b236.pain out=painwide(drop= _NAME_) prefix=l2paintol;


by id cs treatment;


id trial;


var l2paintol;


run;

data painwide1;


set painwide;

*AVERAGE OF FIRST THREE LOG RESPONSES;
av3l2paintol= mean(l2paintol1,l2paintol2,l2paintol3);

run;

proc ttest data=painwide1;


class cs;


var av3l2paintol;


title 'T-Test of Average over Baseline Log2(Paintol) Measurements';


run;
c. Cases with missing observations deleted

How does this test compare to the previous results?

*CASES WITH MISSING VALUES DELETED;
*DELETE ID's WITH MISSING VALUES FROM WIDE DATA SET;
data painwide2;


set painwide1;


if l2paintol1=. or l2paintol2=. or l2paintol3=. or l2paintol4=. then delete;

run;

proc ttest data=painwide2;


class cs;


var av3l2paintol;


title 'T-Test of Average over Baseline Log2(Paintol) Measurements - DELETE MISSING';


run;

2. Paired t-test

d. Average baselines and repeated separately by baseline
i. Overall 

proc ttest data=painwide2;

paired l2paintol4*(av3l2paintol l2paintol1-l2paintol3);

title 'Paired Ttests Overall';

run;

ii. by tmt*coping style

proc sort data=painwide2;by cs treatment ;run;

proc ttest data=painwide2;

by cs treatment ;

paired l2paintol4*(av3l2paintol l2paintol1-l2paintol3);

title 'Paired TTESTS by Coping Style and Treatment';

ods output statistics=statistics ttests=ttests;

run;

*MAKE TABLE TO LOOK LIKE PAGE 92 OF TEXT;
*MERGE OUTPUT DATASETS AND PRINT TABLE;
data both;

merge statistics ttests;

by cs treatment;

run;

proc print data=both;

where difference='l2paintol4 - l2paintol2';

var cs treatment n mean stddev stderr tvalue probt;

run;

3. Difference of differences (4th observation minus 2nd observation)

What do you note about these results?
*DIFFERENCE OF DIFFERENCES, using 58 complete cases;
data diff;


set painwide2;


*(4th observation - 2nd observation);

difvar=l2paintol4-l2paintol2;

run;

proc sort data=diff; by cs treatment;run;

e. By coping style:

i. Compare attend tmt to null tmt

proc ttest data=diff;

where treatment~='distract' and cs~='distracter';

class treatment;

var difvar;

title 'Difference of Differences TTEST for ATTENDERS';

run;

ii. Compare distract tmt to null tmt

proc ttest data=diff;

where treatment~='attend' and cs~='attender';

class treatment;

var difvar;

title 'Difference of Differences TTEST for DISTRACTERS';

run;
4. Random intercept model

We model the coping style main effect only at this time.  

TITLE1 'log pain tol';

proc mixed data=b236.pain method=reml covtest cl;

class id cs;

model l2paintol=cs /  s cl;

estimate 'attenders' intercept 1 cs 1 0 / cl;

random intercept / type=un subject=id;

run;

TODO:  Find the following sections of the output and try to figure out what information is presented there:  
                               Model Information
                             Class Level Information
                                   Dimensions
                                Iteration History
                           Convergence criteria met
                         Covariance Parameter Estimates
                                 Fit Statistics
                        Null Model Likelihood Ratio Test
                          Solution for Fixed Effects
                       Type 3 Tests of Fixed Effects
            
Also look in the log to check on convergence. This is always important to inspect. 
Is CS Significant? What two places in the output reveal this?

